
; ==================================================================

;

;   Simulation of an laterally loaded pile - full gap

;

; ==================================================================

set logfile pl2.log

set log on

;

new

title

Laterally Loaded Pile Demonstration Problem

;

; =======================================================

; Create a single soil block, set its material properties,

; specify boundary conditions, apply in-situ stresses,

; turn on gravity, and ensure that the gravity forces are

; in balance with the in-situ stresses.

;

gen zone brick size 7 7 7 edge=11

;

mod elastic

pro bulk=5e9 she=1e9 density=2000

;

fix z range z=(-0.01, 0.01)

fix x range x=(-0.01, 0.01)

fix x range x=(10.99, 11.01)

fix y range y=(-0.01, 0.01)

fix y range y=(10.99, 11.01)

;

init szz -2.20e5 grad=(0, 0, 2.0e4)

init sxx -1.32e5 grad=(0, 0, 1.2e4)

init syy -0.88e5 grad=(0, 0, 8.0e3)

;

set gravity=(0.0, 0.0, -10.0)

solve

;

; =======================================================

; Create a pile in the center of the soil block; set the

; properties of the pile.

;

sel pile id=1 begin=(5.5, 5.5, 12.0) end=(5.5, 5.5, 4.0) nseg=8

;

sel pile prop Emod=8.0e10 Nu=0.30 XCArea=0.7854 &

XCJ=9.82e-2 XCIy=4.91e-2 XCIz=4.91e-2 &



Per=3.14 &

              CS_sK=1.3e11 CS_sCoh=1.0e10 CS_sFric=30.0 &

              CS_nK=1.3e09 CS_nCoh=1.0e04 CS_nFric=0.0 CS_nGap=on

;

; =======================================================

; Set up histories for monitoring model behavior

;

history nstep=10

hist id=1 unbal

hist id=20 sel node xdisp id=1

;

hist id=100 sel pileSEL Fz id=1

hist id=200 sel pileSEL Fz id=3

hist id=300 sel pileSEL Fz id=5

hist id=400 sel pileSEL Fz id=7

hist id=500 sel pileSEL Fz id=8

;

hist id=101 sel pileSEL My id=1

hist id=201 sel pileSEL My id=3

hist id=301 sel pileSEL My id=5

hist id=401 sel pileSEL My id=7

hist id=501 sel pileSEL My id=8

;

; =======================================================

; Apply the displacements (three phases: forward, back, forward)

;

;

def initval

  push_vel = 1.0e-8

push_velm = -push_vel

end

initval

;

cycle 1  ; force the local-nodal systems of all nodes associated with

         ; pile to be updated

sel node FIX  x          range cid=1

sel node INIT xVel=push_vel  range cid=1

cycle 40000

sav pl2_1.sav

;

sel node INIT xVel=push_velm  range cid=1

cycle 80000

sav pl2_2.sav

;



sel node INIT xVel=push_vel  range cid=1

cycle 40000

sav pl2_3.sav

;

; =======================================================

; Turn off logfile

;

set log off

;

; =======================================================

return

�


